

# CONTENTS

---

<b>BoxPlot ヘルプ</b> .....	2
AddWavesToBoxPlot コマンドのヘルプ .....	2
AppendBoxPlot コマンドのヘルプ .....	3
サンプル1 (垂直ボックスプロット) .....	4
サンプル2 (水平ボックスプロット) .....	5
ModifyBoxPlot コマンドのヘルプ .....	5
サンプル3 (ポイントごとにマーカー表示を設定) .....	13
サンプル4 (マーカーの色や大きさを設定) .....	15

# BoxPlot ヘルプ

※Igor Pro PDF マニュアルの記載から掘り下げて使いたい場合は、このヘルプを参照してください。

また、Example Experiment は、古い実装方法で記載されているため、新しいバージョンを使っている方は、こちらを参照してください。

ここで説明するコマンドは、Igor Pro 8.0 以降で使うことができます。

## AddWavesToBoxPlot コマンドのヘルプ

**AddWavesToBoxPlot** [/W=*winName* /T=*traceName* /INST=*traceInstance*] *wave* [, *wave*] ...

AppendBoxPlot コマンドによって作成された既存のボックスプロットトレースに、1D ウェーブを追加します。

AddWavesToBoxPlot は Igor Pro 8.0 で追加されました。

ボックスプロットのトレースでは、トレース内の各データセットを定義するために多数のウェーブが必要になる場合があります。また、ウェーブ名が非常に長くなる場合があるため、AppendBoxPlot で開始されたリストにウェーブを追加する AddWavesToBoxPlot コマンドを提供しています。

### フラグ

*/T=traceName*

*/INST=traceInstance*

これらのフラグは、ウェーブが追加される既存のボックスプロットトレースの名前とインスタンス番号を指定します。

*/INST* を指定せずに */T* を使うこともできます。

その場合、インスタンス番号が 0 のトレースが使われます。

*/T* なしでは */INST* を使わないでください。

トレース名とトレースインスタンス番号については、ヘルプ Graphs.ihf の「Creating Graphs」を参照してください。

*/T* と */INST* の両方が指定されていない場合、グラフ上の最前面のボックスプロットのトレースがデフォルトで使われます。

通常、最も新しく追加されたボックスプロットのトレースになります。

*/W=winName*

指定したグラフウィンドウまたはサブウィンドウに追加します。

省略された場合、AddWavesToBoxPlot はアクティブウィンドウまたはサブウィンドウ上で動作します。

*winName* でサブウィンドウを識別する時には、ウィンドウ階層を構成する方法の詳細を参照してください。

### 詳細

元の AppendBoxPlot コマンドに X ウェーブが含まれていた場合、ボックスプロットデータセットのウェーブのリスト内のウェーブの総数は、X ウェーブのポイント数を超えることはできません。

ボックスプロットのトレースが複数列のウェーブで定義されている場合、このコマンドでウェーブを追加することはできません。

## AppendBoxPlot コマンドのヘルプ

**AppendBoxPlot** [*axis flags*] [/W=*winName* /TN=*traceName* /VERT[=*doVert*] /CATL[=*doCatLabels*]]  
wave[, wave, ...] [vs *xWave*]

AppendBoxPlot コマンドは、ボックスプロットのトレースをターゲットまたは指定されたグラフに追加します。

ボックスプロットは、データ値の分布の概要を表示する方法です。

データの分布の概要を表示する別の方法として、バイオリンプロットがあります。

AppendBoxPlot は Igor Pro 8.0 で追加されました。

Igor では、ボックスプロットトレースは1つのグラフトレースとして扱われ、グラフからの削除やトレースの並べ替えなど、多くの操作は、他のグラフトレースと同様にボックスプロットトレースでも同じように機能します。

DisplayBoxPlot というコマンドはありません。

Display コマンドの後に AppendBoxPlot を使用してください。

### パラメーター

ボックスプロットの1つのボックスのデータは、1つの1D ウェーブ全体、または複数列のウェーブの1つの列から取得されます。

トレース内のボックスプロットの数、ウェーブのリスト内の1D ウェーブの数、または1つの複数列ウェーブの列の数によって決定されます。

1D と複数列のウェーブを混在させることは許可されていません。

1D ウェーブは 100 個まで登録できます。

ボックスプロットトレースで 100 以上のボックスが必要な場合は、複数列のウェーブを使うか、AddWavesToBoxPlot コマンドを使ってリストに追加する必要があります。

xWave を指定しない場合、各ボックスプロットは X=0、1 などの数値軸上に配置されます。

データが複数列のウェーブの場合、デフォルトでは、マトリックスウェーブの X スケーリングから位置決めが行われます。

数値の xWave を指定することで、各ボックスプロットを X 軸上の任意の位置に配置することができます。

テキスト xWave はカテゴリ軸を使ってボックスプロットを表示します。

複数列のウェーブで /CATL フラグを使うと、ディメンションラベルがカテゴリラベルとして使われたカテゴリ X 軸が結果として得られます。

xWave は、1D の Y ウェーブまたは複数列のウェーブの列数と同じ数以上のポイントが必要です。

コマンドラインに対してウェーブのリストが長くなるような場合には、xWave を使うことができます。リストを完成させるために AddWavesToBoxPlot を使います。

### フラグ

/L/R/B/T これらの軸フラグは、AppendToGraph で使われているものと同じです。

/CATL[=*doCatLabels*]

列ディメンションラベルを使い、列ディメンションラベルをカテゴリラベルとして使うカテゴリ軸のボックスプロットを作成します。

doCatLabels は 0 または 1 で、/CATL は /CATL=1 と同等です。

/TN=*traceName* トレースにカスタム名を付けることができます。

これは、異なるデータフォルダーから取得した同じ名前前のウェーブを表示する時に便利

です。

詳細はヘルプ Programming.ihf の「User-Defined Trace Names」を参照してください。

AppendBoxPlot の /TN フラグは、コマンド名の後に通常の位置で指定します。

/VERT[=*doVert*] 個々のボックスプロットを Y 軸に沿って縦に並べます。

*doVert* は 0 または 1 で、/VERT は /VERT=1 と同じです。

/VERT は、トレースごとに ModifyGraph SwapXY と似た処理を行います。

水平ボックスのボックスプロットを作成するには、ModifyGraph SwapXY または AppendBoxPlot/VERT を使います。

/W=*winName* 指定したグラフウィンドウまたはサブウィンドウに追加します。

省略された場合、AddWavesToBoxPlot はアクティブウィンドウまたはサブウィンドウ上で動作します。

Proc、マクロ、コマンドラインで使う場合は、最初に指定する必要があります。

*winName* でサブウィンドウを識別する時には、ウィンドウ階層を構成する方法の詳細を参照してください。

## 詳細

ModifyGraph を使って、ボックスプロットトレースの全体的な外観の一部を設定することができます。デフォルトでは、線の色は黒で、外れ値ではないデータポイントのマーカは、通常のトレースマーカの 0.7 倍の大きさの空洞の円です。

Modify Box Plot ダイアログまたは ModifyBoxPlot コマンドを使うと、通常のトレースと同様に、ModifyGraph で設定したトレース色、線幅、線種、マーカ、マーカサイズを使うように選択できます。

ModifyBoxPlot により、ボックスプロットのトレースのさまざまな部分について、これらの特性を詳細にコントロールすることができます。

外れ値、極端な外れ値、および非外れ値を表すマーカは、ModifyGraph で設定されたマーカを上書きします。

## サンプル 1（垂直ボックスプロット）

新しい Experiment を作成して実行します。

コマンドウィンドウで次を実行します。

```
Make/O/N=(25,3) multicol // 25 行 x3 列のウェーブを作成
SetRandomSeed(.4)
multicol = gnoise(1) // 正規分布に従う 3 つのデータセットを代入
multicol[20][1] = 5 // 「極端な」外れ値
multicol[13][2] = -4 // 外れ値
Make/O/N=3/T labels // カテゴリプロットを作るためのテキストウェーブ
labels = "Dataset #" + num2str(p) // カテゴリプロットの x 軸用のラベル
Display; AppendBoxPlot multicol vs labels
```

結果は次のようになります。


## サンプル2（水平ボックスプロット）

新しい Experiment を作成して実行します。

コマンドウィンドウで次を実行します。

```
Make/O/N=50 ds1, ds2, ds3, ds4 // 4つの1Dウェーブを作成
Make/O/N=4 dsX
ds1 = gnoise(1)
ds2 = gnoise(2)
ds3 = logNormalNoise(0, 1)
ds4 = enoise(2)
dsX = p^2
Display; AppendBoxPlot ds1,ds2,ds3,ds4 vs dsX
ModifyGraph swapXY=1 // 縦横を変換
ModifyGraph margin(top)=20 // 上部のマージンを調整
```

結果は次のようになります。



## ModifyBoxPlot コマンドのヘルプ

**ModifyBoxPlot** [/W=winName] [keyword=value, keyword=value, ...]

ModifyBoxPlot コマンドは、ターゲットまたは指定されたグラフ内のボックスプロットのトレースを変更します。

ボックスプロットトレースを作成するには、AppendBoxPlot を参照してください。

ボックスプロットおよびボックスプロットの各要素の詳細については、ヘルプ Graphs.ihf の「Box Plot」を参照してください。

ModifyBoxPlot は Igor Pro 8.0 で追加されました。

### パラメーター

ModifyBoxPlot のパラメーターは、キーワードと値のペアで構成されています。

trace キーワードは、その後に続くキーワードで対象とするトレースを指定します。


例えば、コマンド：

```
ModifyBoxPlot trace=trace0, boxFill=(49151,60031,65535) // Light blue
```

は、trace0 トレースのすべてのデータセットのボックスファイルを水色に設定します。

前のページのサンプル2に続いて、次のコマンドを入力すると、右のようになります。

```
ModifyBoxPlot boxFill=(49151,60031,65535)
```




Igor Pro 9.0 以降では、キーワードの後にゼロベースのデータセットインデックスを角括弧内に追加することで、特定のトレースの特定のデータセットの設定を変更することができます。

例えば：

```
ModifyBoxPlot trace=trace0, boxFill[1]=(49151,65535,49151) // Light green
```

前のコマンドに続いて、次のコマンドを入力すると、右のようになります。

```
ModifyBoxPlot boxFill[1]=(49151,65535,49151)
```



これにより、2番目のデータセット（インデックス=1）のボックス塗りつぶしが薄緑色に設定され、他のデータセットのボックス塗りつぶしは変更されません。

## 一般的なパラメーター

- Trace=traceName** 変更するボックスプロットトレースの名前を指定します。指定したトレースがボックスプロットトレースでない場合、エラーが発生します。  
trace キーワードを指定しない場合、ModifyBoxPlot はそれがボックスプロットのトレースであるかどうかに関わらず、グラフの最初のトレースを使います。  
ただし、instance キーワードについては例外があります。
- Instance=instanceNum** trace と instance の組み合わせは、ModifyGraph コマンド（ヘルプファイル Igor Reference.ihf）のキーワード (traceName#instanceNum) と同じように動作します。  
trace キーワードを指定しない instance キーワードは、ModifyGraph のキーワードで使われる [traceNum] と同様に、グラフ内の instanceNum 番目のトレースにアクセスします。  
ヘルプ Trace Name (Graphs.ihf) と Object Indexing (Commands.ihf) を参照してください。
- medianIsMarker[=v]** v が省略された場合、または v がゼロ以外の場合、中央値はボックスプロットの箱の部分に線ではなくマーカーで表示されます。
- notched[=n]** n が省略された場合、または 0 以外の場合、ノッチ付きボックスプロットが描画されます。  
ノッチは中央値の 95 パーセント信頼限界を表しています。
- outlierMethod=m**  
**outlierMethod={m, p1, p2, p3, p4}** 生のデータの値がボックスプロット上に描画されると、それらは正常データポイント、外れ値、極端な外れ値として分類されます。

値を分類する方法は 4 つあります。

方法 0 と 1 ではパラメーターは必要なく、outlierMethod=*m* というフォーマットを使うことができます。

方法 2 と 3 では、追加のパラメーターが必要となり、中括弧を使う拡張形式が必要です。

*m*=0 : Tukey 法 (デフォルト) : 外れ値は内側のフェンスの外側の値であり、極端な外れ値は外側のフェンスの外側の値です。  
フェンスの定義など、詳細についてはヘルプ Box Plots (Graphs.ihf) を参照してください。

*m*=1 : 外れ値とは、ウィスカーの両端を越えたデータ値のことです。

*m*=2 : 外れ値とは、平均値 + - 因子 1\*SD を超える値を指します。  
極端な外れ値とは、平均値 + - 因子 2\*SD を超える値を指し、SD は標準偏差です。  
factor1 は p1 によって与えられ、factor2 は p2 によって与えられます。  
パラメーター p3 と p4 は使われません。

*m*=3 : 外れ値および極端な外れ値は、パラメーターによって完全に指定されます。  
p1 は下限の極端な外れ値、p2 は下限の外れ値、p3 は上限の外れ値、p4 は上限の極端な外れ値の境界を設定します。  
4 つのパラメーターすべてが必要です。

outlierMethod キーワードは、Igor Pro 9.0 で追加されました。

quartileMethod=*m* 四分位数の計算に使う手法を選択します (箱の上下) :

*m*=0 : Tukey 法 (デフォルト)  
*m*=1 : Minitab の方法  
*m*=2 : Moore and McCabe 法  
*m*=3 : Mendenhall and Sincich 法

詳細は、StatsQuantiles /QM フラグの説明を参照してください。

showData=*whatData* トレース内の各ボックスプロットの生のデータのサブセットを選択し、マーカーを使って表示します。

*whatData* の値は、以下の名前です :

All : 全てのデータポイントを表示 (デフォルト)  
None : データポイントを表示しない  
Outliners : 外れ値と極端な外れ値のみ表示  
FarOutliers : 極端な外れ値のみ表示

外れ値に関する詳細な情報は、ヘルプ Box Plots (Graphs.ihf) を参照してください。

デフォルトでは、通常のデータポイントのマーカーは通常のトレースマーカーの 0.7 倍の大きさの中空の円、外れ値は通常のサイズの塗りつぶされた円、極端な外れ値は通常のサイズの塗りつぶされた四角形で表示されます。

showFences[=*v*] *v* が省略された場合、またはゼロ以外の場合、フェンスはボックスと同じ幅の点線で表示されます。

フェンスに関する説明については、ヘルプ Box Plots (Graphs.ihf) を参照してください。

*v* はデフォルトで 0 に設定されています。

showMean[=v] v が省略された場合、またはゼロ以外の場合、データの平均値がマーカーとして表示されます。

whiskerMethod=m

whiskerMethod={m [, p1, p2]}

ウィスカーは四分位（箱の上下）から、whiskerMethod で決定された極端な値まで描画されます。

方法 0~5 は余分なパラメーターを必要とせず、whiskerMethod=m というフォーマットで指定することができます。

方法 6 と 7 は追加のパラメーターを必要とし、中括弧を使った拡張フォーマットを必要とします。

拡張フォーマットは、Igor Pro 9.0 で追加されました。

m=0 : 極端なデータ値（デフォルト）

m=1 : 内側のフェンス

m=2 : 「隣接する」ポイント - 内側のフェンス内の最後のデータポイント

m=3 : 平均値から 1 標準偏差離れた値

m=4 : 9 パーセンタイルと 91 パーセンタイル

m=5 : 2 パーセンタイルと 98 パーセンタイル

m=6 : 下限値は、p1 で指定されたパーセンタイルまでボックスから引き出されます。

上限値は、p2 で指定されたパーセンタイルまでボックスから引き出されます。

この方法では p1 と p2 の両方が必要です。

Igor Pro 9.0 で追加されました。

m=7 : ウィスカー先端は、平均データ値±標準偏差の何倍かの値で示されず。

係数は p1 で与えられます。

p2 は本方法では使いません。

Igor Pro 9.0 で追加されました。

フェンスとパーセンタイルに関する説明については、ヘルプ Box Plots (Graphs.ihf) を参照してください。

## 表示パラメーター

boxWidth=w

カテゴリではない X 軸の場合、boxWidth は四分位を示すボックスの幅を設定します。

w が 0 以上 1 以下の場合は、プロット矩形の幅に対する分数として扱われます。

w が 1 より大きい場合、単位はポイントです。

カテゴリ X 軸としてボックスプロットが表示される場合、boxWidth は無視されます。

ボックスの幅はカテゴリプロットボックスと同じに設定され、ModifyGraph barGap と catGap の影響を受けます。

新しいカテゴリではないボックスプロットでは、ボックス幅が

boxWidth=(1/2\*n) に設定されています。

ここで、n はトレース上のデータセット（ボックスの数）の数です。

上記の「サンプル 1」（左）に

```
ModifyBoxPlot boxWidth=0.1
```

を適用すると次のようになります（右）。



capWidth=w

オプションとして、capWidth キーワードでコントロールされる幅のキャップ（水平線）でウィスカーを終端させることができます。

w が 0 から 1 の間であれば、ボックス幅の分数です。


w が 1 より大きい場合、単位はポイントです。

w が 0（デフォルト）の場合、キャップは描画されません。

上記の「サンプル1」（左）に

ModifyBoxPlot capWidth=0.3

を適用すると次のようになります（右）。



jitter=j

密集したデータセットを見やすくするために、表示された各データポイントに水平オフセットを適用します。

引数 j は、ボックス幅の分数として表され、任意のデータポイントに適用される最大オフセットをコントロールします。

j の値は 1 より大きい値でも構いませんが、一般的には 1 未満の値の方がより見栄えがよくなります。

デフォルトは 0.7 です。

lineStyles={boxStyle, whiskerStyle, medianStyle[, capStyle]}

ボックス、ウィスカー、中央線、およびオプションでウィスカーキャップを描画する線種を設定します。

線種についてはヘルプ Dashed Line (Graphs.ihf) を参照してください。

線種 -1 は、ModifyGraph lstyle で設定された線種を使います。

すべてのパラメーターのデフォルトは -1 です。

lineThickness={boxThickness, whiskerThickness, medianThickness[, capThickness]}

ボックス、ウィスカー、中央線、およびオプションでウィスカーキャップを描画する線の太さを設定します。

ゼロより小さい太さは、ModifyGraph lsize キーワードで設定された太さを使用します。

太さがゼロの場合、ボックスプロットの対応する要素が隠されます。

すべてのパラメーターのデフォルトは -1 です。

markers={dataMarker, outlierMarker, farOutlierMarker[, medianMarker, meanMarker]}

通常のデータポイント、外れ値、極端な外れ値、キーワード medianIsMarker が指定された場合は中央値、キーワード showMean が指定された場合は平均値に使用するマーカー番号を設定します。

0	1	2	3	4	5	6	7
○	—		田	⊠	□	✕	✕
8	9	10	11	12	13	14	15
■	▲	◆	●	/	\	▽	▼
16	17	18	19	20	21	22	23
▽	◇	◆	◇	◇	◇	◇	△
24	25	26	27	28	29	30	31
▲	△	▲	▽	▼	▽	▼	#
32	33	34	35	36	37	38	39
◇	○	⊕	⊗	△	◁	◀	◁
40	41	42	43	44	45	46	47
▷	▶	▷	◊	◆	◊	◊	◆
48	49	50	51	52	53	54	55
◊	▲	▲	◇	◆	◊	◊	
56	57	58	59	60	61	62	


これらのパラメーターのデフォルト値は、markers={8, 19, 16, 26, 27} (中空の円、塗りつぶしの円、塗りつぶしの四角形、X、横長の菱形) です。

パラメーターのいずれかを -1 に設定すると、対応するデフォルトのマーカーが使われます。

上記の「サンプル1」(左)に

ModifyBoxPlot markers={2, 15, 50}

を適用すると次のようになります(右)。



markersOnTop={dataOnTop, medianOnTop, meanOnTop}

デフォルトでは、データマーカー、medianIsMarker で有効化されている場合は中央値マーカー、showMean で有効化されている場合は平均値マーカーは、ボックスとウィスカーの線の下に描画され、線が隠れないようにします。しかし、特殊効果によってはマーカーが上部にくる必要があります。

dataOnTop を 1 に設定すると、すべての生のデータポイント(通常データポイント、外れ値、極端な外れ値)がボックス・ウィスカーの上に描画されます。

medianOnTop と meanOnTop は同様に、中央値と平均のマーカーの描画をコントロールします。

markerOnTop キーワードは、Igor Pro 9.0 で新しく追加されました。

markerSizes={dataSize, outlierSize, farOutlierSize[, medianSize, meanSize]}

通常のデータポイント、外れ値、極端な外れ値、キーワード medianIsMarker が指定されている場合は中央値、キーワード showMean が指定されている場合は平均値のマーカサイズを設定します。

マーカサイズがゼロの場合、ModifyGraph で設定されたマーカサイズにスケール係数を乗じた値が使われます。


スケール係数は、外れ値以外のポイントでは 2/3、外れ値と中央値および平均マーカでは 1、極端な外れ値では 4/3 です。

すべてのパラメーターのデフォルト値は 0 です。

上記の「サンプル 1」 (左) に

ModifyBoxPlot markerSizes={5, 4, 3}

を適用すると次のようになります (右)。



opaqueMarkers={opaqueData, opaqueOutliers, opaqueFarOutliers, opaqueMedian, opaqueMean}

中空マーカの内側を不透明に描画し、その下のアイテムを覆い隠します。

このキーワードには、通常のデータポイント、外れ値、極端な外れ値、中央値マーカ (medianIsMarker が有効になっている場合)、平均マーカ (showMean が有効になっている場合) のそれぞれに対して、個別の設定があります。

対応するパラメーターが 1 の場合、指定されたタイプのマーカは不透明に描画されます。

opaqueMarkers キーワードは、Igor Pro 9.0 で新しく追加されました。

## 色のパラメーター

すべての色は (r,g,b[,a]) RGBA 値で指定されます。

color=(0,0,0,0) を指定すると、ModifyGraph rgb で設定された色が使われます。

boxColor=(r,g,b[,a])      ボックス部分の輪郭色を設定します。  
デフォルトの色は黒です。

boxFill=(r,g,b[,a])      ボックスの塗りつぶし色を設定します。

capColor=(r,g,b[,a])      キャップラインがある場合、その色を設定します。  
デフォルトの色は黒です。

dataColor=(r,g,b[,a])      外れ値ではないデータポイントの色を設定します。  
デフォルトの色はトレースの色です。

`dataStrokeColor=(r,g,b[,a])`

外れ値ではないデータポイントのストロークの色を設定します。  
デフォルトの色は黒です。

`dataStrokeColor` キーワードは、Igor Pro 9.0 で追加されました。

`dataFillColor=(r,g,b[,a])` マーカーが中空の場合、外れ値以外のデータポイントの塗りつぶし色を設定します。

デフォルトの色は白です。

`dataFillColor` キーワードは、Igor Pro 9.0 で追加されました。

`farOutlierColor=(r,g,b[,a])`

極端な外れ値のポイントの色を設定します。  
デフォルトの色はトレースの色です。

`farOutlierStrokeColor=(r,g,b[,a])`

極端な外れ値データポイントのストローク色を設定します。  
デフォルトの色は黒です。

`farOutlierStrokeColor` キーワードは、Igor Pro 9.0 で追加されました。

`farOutlierFillColor=(r,g,b[,a])`

マーカーが中空の場合に、極端な外れ値データポイントの塗りつぶし色を設定します。  
デフォルトの色は白です。

`farOutlierFillColor` キーワードは、Igor Pro 9.0 で追加されました。

`meanColor=(r,g,b[,a])` 平均マーカーの色を設定します。

デフォルトの色はトレースの色です。

`meanStrokeColor=(r,g,b[,a])`

`showMean` が有効になっている場合、平均マーカーのストロークの色を設定します。  
デフォルトの色は黒です。

`meanStrokeColor` キーワードは、Igor Pro 9.0 で追加されました。

`meanFillColor=(r,g,b[,a])`

`showMean` が有効になっている場合、平均マーカーの塗りつぶし色を設定します。

マーカーが中空の場合に適用されます。

デフォルトの色は白です。

`meanFillColor` キーワードは、Igor Pro 9.0 で追加されました。

`medianLineColor=(r,g,b[,a])`

中央線の色を設定します。

`medianIsMarker` が設定されていない場合のみ、効果を確認できます。

デフォルトの色は黒です。

`medianMarkerColor=(r,g,b[,a])`

中央マーカーの色を設定します。

`medianIsMarker` が設定されている場合のみ、効果を確認できます。

デフォルトの色はトレースの色です。

`medianStrokeColor=(r,g,b[,a])`

`medianIsMarker` が有効になっている場合に表示される中央値マーカーの線の

色を設定します。  
デフォルトの色は黒です。

medianStrokeColor キーワードは、Igor Pro 9.0 で追加されました。

medianFillColor=(r,g,b[,a])

中央値マーカーの塗りつぶし色を設定します。  
medianIsMarker マーカーが有効になっている場合に表示されます。  
マーカーが中空の場合に適用されます。デフォルトの色は白です。

medianFillColor キーワードは、Igor Pro 9.0 で追加されました。

outlierColor=(r,g,b[,a]) 通常の外れ値のマーカーの色を設定します。  
デフォルトの色はトレースの色です。

outlierStrokeColor=(r,g,b[,a])

外れ値データポイントのストローク色を設定します。  
デフォルトの色は黒です。

outlierStrokeColor キーワードは、Igor Pro 9.0 で追加されました。

outlierFillColor=(r,g,b[,a])

マーカーが中空の場合の外れ値データポイントの塗りつぶし色を設定します。  
デフォルトの色は白です。

outlierFillColor キーワードは、Igor Pro 9.0 で追加されました。

whiskerColor=(r,g,b[,a])

ウィスカーラインの色を設定します。  
デフォルトの色は黒です。

### ボックスプロットのデータポイントごとのマーカー設定

データポイントマーカーの色、マーカーのスタイル、マーカーのサイズの基本設定は、データポイントごとの設定を含むマーカー設定ウェーブを使用して上書きすることができます。  
この機能は、Igor Pro 9.0 で追加されました。

データポイントごとのマーカー設定は、トレース全体（トレースを構成するすべてのデータセット）またはトレースの特定のデータセットに適用できます。

### サンプル3（ポイントごとにマーカー表示を設定）

```
Function DemoBoxPlotPerPointMarkerSettings ()
  Make/O box0={1,2,3,4,5}, box1={2,3,4,5,6}, box2={3,4,5,6,7}
  String title = "Box Plot Per Point Marker Settings"
  Display/W=(557,99,948,310)/N=BoxPlotPerPointPlot as title

  // 3つのデータセット (box0、box1、box2) を含む trace0 というトレースを作成
  AppendBoxPlot/TN=trace0 box0,box1,box2

  // トレース trace0 の全データセットにマーカーとマーカーサイズを設定
  ModifyBoxPlot trace=trace0, markers={18,-1,-1} // 18=Diamond
  ModifyBoxPlot trace=trace0, markerSizes={5,5,5}

  // trace0 のすべてのデータセットに対して、データポイントごとのマーカーを設定
  Make/O boxMarkers = {15,16,17,18,19}
  ModifyBoxPlot trace=trace0, dataMarkerWave=boxMarkers
```

```
// データセット box1 のみにデータポイントごとのマーカーを設定
Make/O boxMarkersForBox1 = {32,33,34,35,36}
ModifyBoxPlot trace=trace0, dataMarkerWave[1]=boxMarkersForBox1
```


End

新しい Experiment を作成し、Procedure Window に上記のコードを貼り付け、コンパイルします。

コマンドウィンドウで

DemoBoxPlotPerPointMarkerSettings ()

を実行すると、次のようにさまざまなマーカーが表示されたプロットになります。



通常、マーカー設定のウェーブは、指定されたデータセット内のデータポイントの数と同じ行数を持ちますが、必ずしもそうである必要はありません。

設定ウェーブの行数がデータセットよりも少ない場合、余分なデータポイントは基本設定を維持します。設定ウェーブの行数がデータセットの行数よりも多い場合、余分な設定ウェーブポイントは使用されません。

詳細および例については、ヘルプ Making Each Data Point Look Different (Graphs.ihf) を参照してください。

dataColorWave [=colorWave]

colorWave を設定して、データポイントマーカーの色をポイントごとに上書きします。

ウェーブは、赤、緑、青、オプションでアルファ値を含む 3 列または 4 列のウェーブである必要があります。

「=colorWave」を省略すると、それまでのマーカーの色ウェーブ設定はクリアされます。

dataColorWave キーワードは、Igor Pro 9.0 で追加されました。

dataMarkerWave [=markerWave]

markerWave を設定して、データポイントマーカーをポイントごとに上書きします。

マーカーウェーブの値は、標準グラフマーカーの番号です。

+	×	*	⊗	⊗	□	△	◇
0	1	2	3	4	5	6	7
○	—		田	⊗	□	⊗	⊗
8	9	10	11	12	13	14	15
■	▲	◆	●	/	\	▽	▼
16	17	18	19	20	21	22	23
▽	◇	◆	◇	◇	◆	◇	△
24	25	26	27	28	29	30	31
▲	△	▲	▽	▼	▽	▼	#
32	33	34	35	36	37	38	39
◇	○	⊕	⊗	△	◁	◀	◁
40	41	42	43	44	45	46	47
▷	▶	▷	◇	◆	◇	◇	◆
48	49	50	51	52	53	54	55
◇	▲	▲	◇	◆	◇	◆	
56	57	58	59	60	61	62	

マーカーと関連マーカー番号の詳細については、ヘルプ Markers (Graphs.ihf) を参照してください。

マーカー番号は有効な範囲にクリップされます。

「=*markerWave*」を省略すると、それまでのマーカーウェーブの設定はクリアされます。

*dataMarkerWave* キーワードは、Igor Pro 9.0 で追加されました。

*dataSizeWave* [=*markerSizeWave*]

マーカーサイズウェーブを設定して、データポイントのマーカーサイズをポイントごとに上書きします。

マーカーウェーブの値は [0,200] の範囲にクリップされます。

「=*markerSizeWave*」を省略すると、それまでのマーカーサイズウェーブの設定はクリアされます。

*dataSizeWave* キーワードは、Igor Pro 9.0 で追加されました。

#### サンプル4 (マーカーの色や大きさを設定)

```

Make/O/N=(25,3) multicol // 25行3列のウェーブ
SetRandomSeed(.4) // 再現可能な乱数
multicol = gnoise(1) // 正規分布に従う3つのデータセット
multicol[20][1] = 5 // 極端な外れ値
multicol[13][2] = -4 // 外れ値
Display; AppendBoxPlot multicol

ModifyGraph lSize=2
ModifyBoxPlot markers={8,19,19}
ModifyBoxPlot markerSizes={3,5,9}
ModifyBoxPlot capWidth=0.5
ModifyBoxPlot boxColor=(0,0,65535)
ModifyBoxPlot medianLineColor=(0,0,65535)


```

```

ModifyBoxPlot whiskerColor=(40000,40000,65535)
ModifyBoxPlot capColor=(40000,40000,65535)
ModifyBoxPlot boxFill=(0,0,65535,20000)
ModifyBoxPlot dataColor=(0,0,0)
ModifyBoxPlot outlierColor=(0,0,0)
ModifyBoxPlot farOutlierColor=(0,0,0)
ModifyBoxPlot jitter=0.75

```

新しい Experiment を作成して、これを実行すると、次のようになります。




さらに続いて次の2行を実行するとマーカーの表示が変わります。

```

ModifyBoxPlot trace=multicol,markerThick={1,0,0},markersFilled={1,0,0,0,0}
ModifyBoxPlot trace=multicol,markerThick={1,0,0},dataFillColor=(2,39321,1)

```



中央のボックスプロットのマーカーの色を変えるには次を実行します。

```

ModifyBoxPlot trace=multicol,dataFillColor[1]=(0,65535,0)

```

